
Rationale
    

I've written this rationale for people who are already familiar with
Elixir's built-in Access module and its related functions
(get_in/2, put_in/3, update_in/3, etc.) If you're not familiar
with Access, you might want to read "For Access novices", then
skip to Part 2 on this page.

  
    
  
  Part 1: Why not Access?


Lenses are a tool for working with nested data structures, which I'll
call containers. Elixir already comes with a tool for that:
 iex> container = %{top_level: [%{lower: 3}, %{lower: 4}]}
 iex> put_in(container, [:top_level, Access.all, :lower], :NEW)
 %{top_level: [%{lower: :NEW}, %{lower: :NEW}]}
Lenses provide an alternative:
 iex> container = %{top_level: [%{lower: 3}, %{lower: 4}]}  # same as above
 iex> lens = Lens.key(:top_level) |> Lens.all |> Lens.key(:lower)
 iex> Deeply.put(container, lens, :NEW)
 %{top_level: [%{lower: :NEW}, %{lower: :NEW}]}             # same as above
That looks like a lot of extra characters to achieve the same result. So: why lenses?
	Lenses have more power. That is, there are more specialized functions like
Access.all/0. Things that are awkward or impossible with just Access
are built in:
# Increment all map values:
iex> Deeply.update(%{a: 1, b: 2, c: 3}, Lens.map_values, & &1+1)
%{c: 4, a: 2, b: 3}

	Lenses can work with types that don't implement the Access behaviour. For example,
MapSets don't have keys, so the MapSet module doesn't implement Access.fetch/2, so you can't
use put_in/3. But you can use Deeply.put when part of your nested container is a MapSet:
iex> container = %{a: MapSet.new([%{aa: 1, bb: 2}])}
%{a: MapSet.new([%{bb: 2, aa: 1}])}
iex> lens = Lens.key(:a) |> Lens.MapSet.all |> Lens.map_values
iex> Deeply.put(container, lens, :NEW)
%{a: MapSet.new([%{bb: :NEW, aa: :NEW}])}


In a way, what you just read is a lie. The list given to a function
like get_in is a description of how to navigate into a nested data
structure. An element in the list can be any function that has this interface:
  fn
    :get, container, continuation ->
       ...
    :get_and_update, container, get_and_update_function ->
       ...
  end
So you could write functions that obey that interface and descend just
fine into MapSets. One way to look at lenses is that save you the
trouble of writing such functions:
	You can just use Lens.MapSet.all instead of writing it yourself.

	If you write a lot of such functions, you'll find that you're
repeating yourself. The traditional functional language approach to
repetition is to factor it out into smaller functions that you
compose together in different ways. But those functions have already been written.
They have names like Lens.multiple and Lens.repeatedly.



  
    
  
  Part 2: Why a new package?


There already exist lens packages at

For <code class="inline">Access</code> novices
    

Languages like Elixir don't let you change data structures; instead,
you create a new data structure that's different from the
original. That can be somewhat annoying when you want to change a
single place within a deeply nested structure. From now on, I'm
going to call such structures containers. We're interested in nested
containers.
Here's a simple example. We have Network structure. It has various
fields, one of which is :name_to_cluster. That points to a Map
whose keys are atoms (cluster names) and whose values are Cluster structs. A
Cluster has various fields, one of which, :downstream, holds a
MapSet of atoms (also cluster names). Like this:
[image: alt text coming]
We want to add the value :c to the MapSet associated with the cluster named :a. 
This code works:
    new_cluster =
      network.name_to_cluster[:a]
      |> Map.update!(:downstream, & MapSet.put(&1, :c))

    new_map =
      network.name_to_cluster
      |> Map.put(:a, new_cluster)

    %{network | name_to_cluster: new_map}
Fine, but what I'm doing has two conceptual steps:
	Point at the place you want to change.
	Cause the change.

In the above code, those two steps are buried inside a lot of
bookkeeping code that does the work of constructing each level of the
new nested container. What really matters is highlighted below: the
path into the container and the MapSet.put operation.
[image: alt text coming]
I want to describe the path and the update function, then sit back while the
compiler writes the bookkeeping code for me.
You probably know that Elixir offers the Access behaviour that does
just that. Here's a better implementation of the above:
    update_in(network.name_to_cluster[:gate].downstream,
              & MapSet.put(&1, :some_name))
Looks pretty clear: every word is about either the path or the
update to be done on whatever's at its end. Alternately, we can use
syntax that's a little longer but not as reliant on macro magic to
describe the path:
    path = [Access.key(:name_to_cluster), :gate, Access.key(:downstream)]
    update_in(network, path,
              & MapSet.put(&1, :some_name))
So where do lenses come in?
Well, let's say we want to add :some_name to all the clusters. I
was surprised that you couldn't do that with a single update_in
because I thought Access.all/0 would do it. Something like:
    path = [Access.key(:name_to_cluster),
            Access.all(),    # `all` will produce list of {key, value} tuples.
            Access.elem(1)]  # Take the value
But no joy:
** (RuntimeError) Access.all/0 expected a list, got: %{gate: %Cluster{downstream: MapSet.new([:big_edit, :has_fragments]), name: :gate}, watcher: %Cluster{downstream: MapSet.new([]), name: :watcher}}
To handle multiple clusters, you need some kind of a loop, perhaps
using for. for does what I expected Access.all to do, and
produces tuples:
    new_map =
      for {name, cluster} <- network.name_to_cluster, into: %{} do
        new_cluster = update_in(cluster.downstream, & MapSet.put(&1, :some_name))
        {name, new_cluster}
      end
    %{network | name_to_cluster: new_map}
There are a variety of other looping or recursive styles you could use. (Actually,
that's sort of a problem: it would be better to have one concise
solution than, say, four that are long enough you actually have to
think to write them or, sometimes, read them.)
In this lens package (and others like it), updating multiple
clusters in the network doesn't require a loop:
   lens = Lens.key(:name_to_cluster) |> Lens.map_values |> Lens.key(:downstream)
                                        ^^^^^^^^^^^^^^^
   Deeply.update(network, lens, & MapSet.update(&1, :some_name))
As you'd hope, it looks almost the same to update a subset of the clusters:
   lens = Lens.key(:name_to_cluster) |> Lens.keys([:gate, :watcher]) |> Lens.key(:downstream)
                                        ^^^^^^^^^^^^^^^^^^^^^^^^^^^
   Deeply.update(network, lens, & MapSet.update(&1, :some_name))
Or, if you prefer, you can use update_in/3, because lenses are
functions that implement the behavior that Access requires;
    path = [Access.key(:name_to_cluster),  # Note Access
            Lens.keys([:gate, :watcher]),  # Note Lens
            Access.key(:downstream)]
    update_in(network, path, & MapSet.put(&1, :some_name))
This would be more compelling, I guess, if I were using maps instead
of structs. With maps, you can avoid Access.key:
    path = [:name_to_cluster,
            Lens.keys([:gate, :watcher]),
            :downstream]
    update_in(network, path, & MapSet.put(&1, :some_name))
I personally just use Deeply everywhere except when the
dot.and[bracket].notation fits too well to pass up. I tend to avoid
nested structures that expose the structure of their innards to all
their clients, so that's not as often as you might guess.
You can, however, read this series as descriptions of how to create
custom path elements for update_in and friends (the way Lens.keys
was used above). You can do that directly by writing functions that
match the required Access behaviour, but since lenses are
functions that match that behaviour, maybe learning lenses is still
worthwhile, if only for the greater variety of prepackaged functions.

Are lenses for you?
    


  
    
  
  It depends


Lenses, like Access, are used for getting data from within nested
containers, for putting data in nested containers, and for updating
existing data. They're part of the long tradition of methods to do
CRUD (create/read/update/delete) that includes relational databases,
HTTP verbs, and the like. Whether they're worth learning depends on
you and the work you do.
The out-of-the-box, batteries included, Kernel module functions
get_in/2, put_in/3, update_in/3, and pop_in/2 let you solve some CRUD
problems without writing an annoying amount of bookkeeping code: they allow
short and declarative(ish) solutions. For other CRUD problems, they don't.
Sometimes lenses will work better; they'll let you write straightforward
solutions when update_in/3 and friends would require more convoluted
ones. The question is whether sometimes is often enough. Not in
some abstract sense, but for you (and your team) and for what you do.
How often do you get annoyed writing boilerplate code around nested
data structures? (Or, probably equally important, how often do you not
create a nested structure because working with them is so annoying?)
Is that often enough that it's worth the time to learn enough about lenses to
determine whether they would make your life (and your team's life) better?
(Note: sometimes Access can do things lenses won't. Most notably (to
me), lenses don't support pop_in/2. I'll point out such deficiencies
as this series goes along.)

  
    
  
  A note on choosing


It may seem weird and offputting to say you should decide on
lenses based on your personal and team annoyance with traversing nested
structures in a functional language. Surely I should be saying lenses
are better in some absolute sense? Well...
I got my first job as a programmer in 1975 (at the age of 16). I've
seen a lot of programmers making judgments about technology new to
them. When they reach for generalizations about "all programmers" or
"all programs", they tend to predict poorly. For example, from about 1983
(when I was a Common Lisp implementer) to the early '90s, I heard an
endless number of people say that garbage collection was impractical
for "serious work". Machines (except for
expensive custom hardware)
objectively just weren't fast enough, and would never be fast
enough. Then Java came out, and the conventional wisdom completely
dropped the issue, even for the slow machines of the day. Garbage
collection was now assumed: the question was how to use it most
efficiently, except that most programmers never thought about that at
all. They adopted the new "paradigm" and moved on.)
And don't get me started about the debate (around 1981) of whether C
could ever replace assembly language for serious coding.
In any case: I've seen too many programmer decisions about technology
that are based on personal preference, often driven by what that
programmer is used to, but – and let me emphasize this – presented
as an objective analysis. To that, I repeat the last line of Ernest
Hemingway's
The Sun Also Rises:
"Wouldn't it be pretty to think so?"
I don't think so, so I won't pitch lenses to you as if you were a
dispassionate person optimizing some objective criterion. That is:
	If you are the exception, someone who really does weigh things objectively –
and I believe you might exist, you're just unusual: welcome!
The series will, I hope, give you enough information to rationally
judge. But I'm not going to structure my argument around your
criteria. Sorry!

	If you accept that you are prone to subjective judgments: welcome!
I hope to show you why my subjective judgment about lenses just
might mesh with yours, and that you might find lenses pleasant.



  
    
  
  Efficiency


In my
very crude
benchmarking, lenses take about 1.5X the time of the equivalent
update_in/3 functions and around 2X of get_in/2. That's for a configuration of maps
and structs. For lists, lenses are worse: about 2X update_in and 3X
get_in/2. I imagine you could beat the Access functions with
hand-crafted recursion (especially if you wrote it using
C or
Rust foreign
functions).
The hoary old
adage applies: you're probably better writing the code in the way
most readable to your team, then optimizing after 
benchmarking shows you where the bottlenecks are.
It's worth calling out Pathex, a
lens-like package said to be faster than Access.

Migrating from Lens 1
    

Migration is supposed to be easy.
Note that you can migrate one module at a time.
There are probably gotchas yet to be discovered.
	Put use Lens2 at the top of any module that defines or uses
lenses. 

	Code that makes lenses (Lens.key(:a)) need not be changed.

	Replace the operations like Lens.to_list and Lens.map with their
Lens2.Deeply equivalents. Note that you have to change the
argument order:
 Lens.put(lens, container, value)     # Lens 1
 Deeply.put(container, lens, value)   # Lens 2
Here is a list of changes:
 Lens.each(lens, container, f)         Deeply.each(container, lens, f)
 Lens.get_and_map(lens, container, f)  Deeply.get_and_update(container, lens, f)
 Lens.map(lens, container, f)          Deeply.update(container, lens, f)
 Lens.one!(lens, container)            Deeply.get_only(container, lens) or Deeply.one!(container, lens
 Lens.put(lens, container, v)          Deeply.put(container, lens, v)
 Lens.to_list(lens, container)         Deeply.get_all(container, lens) or Deeply.to_list(container, lens)

	use Lens2 will import macros deflens and deflens_raw, so you
don't have to change them. However, it also imports defmaker and
def_raw_maker, which I prefer.

	If you use
TypedStruct and
TypedStructLens,
you have to
[I don't know what yet. Probably just adding alias Lens2.TypedStructLens to the head of the module].

Lenses move pointers
    

The word "lens" is a metaphor. We're supposed to gain some
understanding of what this particular software construct does, based on our
experience with physical lenses in the real world.
I never found the metaphor helpful when learning lenses. And since lenses
have the reputation of being hard to learn, I'm happy to abandon
it (while keeping the name everyone uses). To me, a more helpful metaphor is the pointer. In the picture below,
you see someone using a slim wooden stick to point to something on a
blackboard. The other picture shows an arrow – a pointer – pointing at a blob
with smaller blobs within it. These are – metaphorically – the same action.
[image: Alt-text is coming]1
The right-hand image is of a data structure with smaller data
structures within it. I'm going to call the bigger structure a
container. The contained structures may also be containers that have
their own internal structure. Or they may be "atomic" values like integers or atoms.
The container has something pointing at it. Perhaps that's a variable
it's bound to via pattern matching:
{:ok, container} = make_that_container(...)
Perhaps the pointer
indicates our container is part of a larger container, and the pointer represents
that you can go from the larger to the smaller, perhaps via constructs like these:
larger.smaller
Map.get(larger, :smaller)
Enum.at(larger, 3)
What a lens does is transform one set of pointers to another. It 
takes a set of pointers to containers as its input. From them, it produces a new set of
pointers, typically to values within the original containers. Like this:
[image: Alt-text is coming]
The lens is the "becomes" in the picture: the creation of the five pointers on the
right from the single pointer on the left. That is, a lens is a
function:
iex> use Lens2     # sets up aliases `Lens` and `Deeply`
iex> Lens.at(3)
#Function<1.126734921/3 in Lens2.Lenses.Indexed.at/1>
... and Lens.at is a function-making (or "higher order")
function. I'm going to give such higher-order functions the name "lens
makers". It's kind of an ugly name, but it's important to remember the
difference between a lens function and the function that makes it.
Except when you're writing a new lens from scratch, your code never
calls a lens function directly. Instead, your code passes the lens to
some other function that does that work. As far as you're concerned,
the lens is no different than the 3 in Enum.at(container, 3) or
the :a in Map.key(container, :a)
To peek ahead, the functions that use lenses are in the Lens2.Deeply package (usually aliased as Deeply). Here's one:
iex> Deeply.update([0, 1, 2], Lens.at(1), & &1 * 1111)
[0, 1111, 2]
I'm going to call such functions operations because they use lenses to operate on containers.

  
    
  
  Lenses are all about "zero, one, many"


Functions like Map.get/3, Enum.at/2, and so on are about a single
value within a container. Their conceptual extension to functions
like get_in/2 share that assumption. There are
exceptions like Access.all/0 or Access.slice/1, but I think it
fair to say those are special cases: both conceptually and in common
usage.
Lenses have a different base assumption: their code expects to be both
consuming and producing multiple pointers: maybe just one, maybe ten,
maybe even zero.
Let's look at some examples.
Here's a picture of a struct or a map or maybe a keyword list.
[image: Alt-text is coming]
The
container contains a set of {key, value} tuples. In the case of a
keyword list, that's the literal, concrete representation. Maps might have
a different internal structure, but the values within a map are always
presented to the outside world as a tuple. Like this:
iex> for elt <- %{a: 1, b: 2}, do: IO.inspect(elt)
{:a, 1}
{:b, 2}
To get a lens that converts a pointer-to-Map into pointers to all the values, you do this:
iex> lens = Lens.map_values
(All my examples implicitly use Lens2, which provides the Lens
alias so that I don't have to write the function's real name,
Lens2.Lenses.Keyed.map_values/0.)
When it's used, the lens will produce these pointers for the container I showed above: 
[image: Alt-text is coming]
You can use that lens to set every element in the container by
wrapping the lens in a singleton list and passing it to
put_in/3:
iex> map = %{a: 1, b: 2, c: 3, d: 4, e: 5}
iex> put_in(map, [lens], :NEW)
                 ^^^^^^
%{c: :NEW, a: :NEW, d: :NEW, e: :NEW, b: :NEW}
     ^^^^     ^^^^     ^^^^     ^^^^     ^^^^
Similarly, you can use the lens with get_in/2 and update_in/3):
iex> get_in(map, [lens])
[3, 1, 4, 5, 2]
iex> update_in(map, [lens], & &1 * 1111)
%{c: 3333, a: 1111, d: 4444, e: 5555, b: 2222}
(I'll note here that Access.all/0 can only be used on lists, so you
can't use out-of-the-box Elixir to do what we just did – except by coding manually the function that Lens2.Lenses.Keyed.map_values/0 gives you for free. 
Having to wrap the lens in a list is a little annoying, so you can use
functions from Lens2.Deeply instead. (There are other reasons to use
them, which I'll cover later.)
Here's Lens2.Deeply.put/3:
iex> Deeply.put(map, lens, :NEW)
%{c: :NEW, a: :NEW, d: :NEW, e: :NEW, b: :NEW}

Lens.map_values/0 points to all the map's values. Unsurprisingly,
you can point to just one value instead:
iex> lens = Lens.key(:c)
[image: Alt-text is coming]
And then you can update only :c's value:
iex> Deeply.update(map, lens, & 1111111*&1)
%{c: 3333333, a: 1, d: 4, e: 5, b: 2}
But let's see what happens if we use this new lens with get_in/2:
iex> get_in(map, [lens])
[3]
That's weird. We wouldn't expect a list from this:
iex> get_in(map, [:c])
3
... so why do we get it with a lens? It's because a lens used with
get_in might point to multiple elements of the container, so it's
always working internally with lists. If it treated singleton lists
differently (by unwrapping them), calling code couldn't distinguish
between a single value that is a list or a list of multiple
values. So everything stays in a list.
To remind you that you always get a list, the Deeply equivalent to
get_in/2 is named Lens2.Deeply.get_all/2. I apologize in advance
for the number of times you'll habitually type Deeply.get(...) and
then be annoyed by this compiler error:
 Lens2.Deeply.get/2 is undefined or private. Did you mean:

       * get_all/2
       * get_only/2
Note Lens2.Deeply.get_only/2. If you're really truly sure you'll get
a singleton list, you can use it to unwrap the value for you. If it's
not a singleton list, get_only will raise an error.

As a final example, you can point at a subset of the values:
[image: Alt-text is coming]
iex> lens = Lens.keys([:a, :e])
iex> Deeply.get_all(map, lens)
[1, 5]

This has almost the same behavior as Map.take/2, except the latter returns a map:
iex> Map.take(map, [:a, :e])
%{a: 1, e: 5}

	&#x21A9;The photo is via Fagnar Brack. Found via a DuckDuckGo 

Pointing into nested containers
    

Lenses' claim to fame is the ability to descend through nested data
structures in a variety of ways, for a variety of structures. This
page is about how you use them to do that.

  
    
  
  Composing lens makers


Lens-creating functions can be combined using the pipe (|>)
macro.
Suppose I've used Lens2 and so can refer to lens-maker functions
with Lens. Here, again, are pictures showing the pointed-at values
from applying, respectively, Lens.key(:c) and Lens.key([:a, :e])
to two copies of the same five-element map.
[image: Alt-text is coming]
However, let's suppose the second map is actually embedded within the first, as the value of the key :c:
[image: Alt-text is coming]
We want the pointers into that nested map, specifically at the values of the :a and :e keys:
[image: Alt-text is coming]
That's easy to do by composing the Lens-making functions:
iex>  use Lens2
iex>  lens = Lens.key(:c) |> Lens.keys([:a, :e])
#Function<13.52599976/3 in Lens2.Lenses.Combine.seq/2>
Let's see the new composed lens at work.
iex>  map = %{a: 1, b: 2, c: 3, d: 4, e: 5}
%{c: 3, a: 1, d: 4, e: 5, b: 2}
iex>  nested = %{map | c: map}
%{c: %{c: 3, a: 1, d: 4, e: 5, b: 2}, a: 1, d: 4, e: 5, b: 2}

iex>  Deeply.get_all(nested, lens)
[1, 5]
iex>  Deeply.put(nested, lens, :NEW)
%{c: %{c: 3, a: :NEW, d: 4, e: :NEW, b: 2}, a: 1, d: 4, e: 5, b: 2}
                ^^^^           ^^^^

  
    
  
  Another example: filtering pointers


Lenses don't have to descend into a data structure. Some lenses can
remove pointers. For example, consider this map from numbers to names:
 iex> map = %{1 => "one", 2 => "two", 3 => "three"}
Lens.map_values will transform a pointer to the map into pointers to the values:
 iex> Deeply.get_all(map, Lens.map_values)
 ["one", "two", "three"]
We can use Lens2.Lenses.Filter.filter/1 (aliased to Lens.filter/1) to make a lens that retains only pointers to three-character names:
 iex> lens = Lens.map_values |> Lens.filter(& String.length(&1) == 3)
... and use it to boost those names' visibility:
iex> Deeply.update(map, lens, & String.upcase(&1))
%{1 => "ONE", 2 => "TWO", 3 => "three"}

  
    
  
  How pipelining works


Normally, we think of a maker like Lens.keys as taking a single argument:
iex> Lens.keys[:a, :b]
However, each lens maker has a two-argument version whose first
argument is a lens (not a lens maker). Therefore this pipeline:
iex> lens = Lens.map_values() |> Lens.keys([:a, :b])
is just syntactic sugar for this:
iex> lens = Lens.keys(Lens.map_values(), [:a, :b])
The second variant doesn't explicitly appear in API documentation
because the documentation for dozens of lens makers would just say
"This works the same as every other lens maker that takes a first
lens argument."

If you're like me, pipelining can trip you up. A few times I've had a lens defined for a particular
data structure, like this one for a two-level map:
iex> two_level_lens = Lens.key(:a) |> Lens.key(:aa)
... and then I have a list containing such maps and want a lens that will pick
out all the :aa values of all the maps in the list:
iex> Deeply.get_all(list_of_maps, a_bigger_lens)
[1, 2]
The question is: how to easily make a_bigger_lens? Because I know
that Lens2.Lenses.Enum.all/0 will give me all elements of a list, a simple
composition should do it:
iex> a_bigger_lens = Lens.all |> two_level_lens.()
But oops:
** (BadArityError) #Function<19.51288540/3 in Lens2.Lenses.Combine.seq/2>
with arity 3 called with 1 argument
(#Function<1.73076862/3 in Lens2.Lenses.Enum.all/0>)
Here I'm passing a lens (constructed by the lens maker Lens.all) to
another lens, not to a lens maker. That is, I'm passing the new lens to the result of Lens.key(:a) |> Lens.key(:aa), not – as I sometimes sloppily assume – to Lens.key/1. 
The way to compose a new lens with an existing lens is to use Lens2.Lenses.Combine.seq/2:
iex> a_bigger_lens = Lens.all |> Lens.seq(two_level_lens)
# or, without the pipeline:
iex> a_bigger_lens = Lens.seq(Lens.all, two_level_lens)
In fact, the definition of the hidden two-argument version of, for
example, Lens2.Lenses.Keyed.keys/1 uses Lens.seq/2:
def keys(previous_lens, key_list) do
  Lens.seq(previous_lens, keys(key_list))
end

  
    
  
  Defining a lens maker


There are two ways to define a lens maker: coding one up from scratch,
or composing existing lens makers. The first way is rare and more
complicated, so I'll put that off and talk only about composition.
Suppose I frequently want to descend two levels into a nested map. Rather than write code like:
Lens.key(:level1_key) |> Lens.key(:level2_key)
... all over the place, I prefer to write a function that does the busywork for me:
MyLens.nested(:level1_key, :level2_key)
I could use a simple def:
def nested(level1, level2),
    do: Lens.key(level1) |> Lens.key(level2)
That works for making an isolated lens, but it doesn't work for composition:
iex> Lens.at(0) |> MyLens.nested(:a, :b)
** MyLens.nested/3 is undefined or private. Did you mean:

       * nested/2    
The problem is that I haven't defined the extra-argument version. I could do that easily enough:
def nested(lens, level1, level2),
  do: Lens.seq(lens, nested(level1, level2)
However, that code will always always look the same, so there's a macro that defines both versions with a single definition:
defmaker nested(level1, level2),
  do: Lens.key(level1) |> Lens.key(level2)
(In the Lens 1 package,
this is called deflens and you can still use that name if you
prefer. But I'm hoping the name will help you
keep in mind that you're not defining a lens but rather a lens
maker, thus saving you from some mistakes. Also: in Lens 1, you
define a lens maker from scratch with deflens_raw. I prefer
def_raw_maker/2.)
You'll probably write a lot of lens makers of this sort.

Missing and nil values
    

In Elixir, a nil sometimes means "there is nothing here" and
sometimes "there is something here, specifically nil.
iex> Map.get(%{a: nil}, :a)
nil
iex> Map.get(%{      }, :a)
nil
Sometimes you want to handle the two cases differently. For example:
iex> Map.put_new(%{a: 1}, :a, :NEW)
%{a: 1}
iex> Map.put_new(%{    }, :a, :NEW)
%{a: :NEW}
This page is how you navigate such distinctions when using lenses. Unlike
Map, where you choose an operation (Map.put/3
vs. Map.put_new/3), with lenses you choose a different lens maker.

  
    
  
  Keyed lenses (maps, structs, and Access.fetch)


Lenses for map structures come in three varieties, such as Lens.key,
Lens.key?, and Lens.key!. (There are also Lens.keys, Lens.keys?, and Lens.keys!.)
Lens.key treats a missing value and nil the same way:
iex> use Lens2
iex> Deeply.get_all(%{a: nil}, Lens.key(:a))
[nil]
iex> Deeply.get_all(%{      }, Lens.key(:a))
[nil]

iex> Deeply.put(%{a: nil}, Lens.key(:a), :NEW)
%{a: :NEW}
iex> Deeply.put(%{      }, Lens.key(:a), :NEW)
%{a: :NEW}

iex> Deeply.update(%{a: nil}, Lens.key(:a), &inspect/1)
%{a: "nil"}
iex> Deeply.update(%{      }, Lens.key(:a), &inspect/1)
%{a: "nil"}
If the container is a struct, Deeply.get_all behaves the same as for
a plain map. It will still produce a nil for a missing key.
iex> Deeply.get_all(%Point{}, Lens.key(:missing))
[nil]
Deeply.put with a missing key is alarming:
iex> Deeply.put(%Point{}, Lens.key(:missing), :NEW)
%{missing: :NEW, y: 2, __struct__: Point, x: 1}
We've destroyed the contract for Point. This is, however, the same thing that put_in/3 would do:
iex> put_in(%Point{x: 1, y: 2}, [Access.key(:missing)], :NEW)
%{missing: :NEW, y: 2, __struct__: Point, x: 1}
I assume there's a reason for that, but I don't know what it is.
Deeply.update can also be used to add fields to a struct, as can update_in/3.

  
    
  
  key?


Lens.key? or Lens.keys?, in contrast, treat nil as a regular value, but handle
a missing value by doing nothing. For Deeply.get_all, nothing is
included in the return list:
iex> Deeply.get_all(%{a: nil, b: 1}, Lens.keys?([:a, :b, :missing]))
[nil, 1] # `keys` would have provided an extra `nil`
Deeply.put will only override an existing value. (It's like a put_not_new.)
iex> Deeply.put(%{a: nil, b: 1}, Lens.keys?([:a, :b, :missing]), :NEW)
%{a: :NEW, b: :NEW}
For Deeply.update, the update function is never called for a missing value.
iex> Deeply.update(%{a: nil, b: 1}, Lens.keys?([:a, :b, :missing]), &inspect/1)
%{a: "nil", b: "1"}
Lens.key? works the same on structs as on plain maps: it cannot create missing values.

  
    
  
  key!


Lens.key! will raise an error whenever it detects any missing key.
iex> Deeply.get_all(%{}, Lens.key!(:missing))
** (KeyError) key :missing not found in: %{}

iex> Deeply.put(%{a: 1}, Lens.keys!([:a, :missing]), :NEW)
** (KeyError) key :missing not found in: %{a: :NEW}

iex(30)> Deeply.put(%{}, Lens.key!(:missing), &inspect/1)
** (KeyError) key :missing not found in: %{}
Structs are handled the same way.

  
    
  
  Other types


Any container that implements the Access behaviour will be treated like a map.
More precisely, 
	Lens.key uses Access.fetch/2 but converts an :error return
value into nil. It uses Access.get_and_update/3 to update (or
put) values.

	Lens.key? uses the same two functions but does nothing in the case
where Access.fetch/2 returns :error.

	Lens.key! does the same, but it raises an error when
Access.fetch/2 returns :error.



  
    
  
  Indexed lenses (lists, tuples)


The core function is Lens2.Lenses.Indexed.at/1. It and its
derivative, Lens2.Lenses.Indexed.indices/1, will return nil when getting an
index that's out of bounds:
  iex> Deeply.get_all([0, 1], Lens.at(2))
  [nil]
  iex> Deeply.get_all([0, 1], Lens.indices([0, 10000]))
  [0, nil]
This is consistent with the behavior of Enum.at/2 and also Lens.key.
When it comes to Deeply.put and Deeply.update, no change is made to an out-of-bound index:
iex> Deeply.put([0, 1], Lens.at(2), :NEW)
[0, 1]
iex> Deeply.update([0, 1], Lens.indices([0, 2]), &inspect/1)
["0", 1]
This is consistent with List.replace_at/3. One annoyance when using
Deeply.update is that a nil (signifying "missing") is passed to the
update function and then the return value is ignored. That's a
problem in the common case when the update function doesn't expect a nil:
 iex> Deeply.update(["0", "1"], Lens.at(2), &Integer.parse/1)
 ** (FunctionClauseError) no function clause matching in Integer.parse/2
     The following arguments were given to Integer.parse/2:

         # 1
         nil
This is not consistent with the behavior of update_in, and I'm inclined to think it a bug. 
iex> update_in(["0", "1"], [Access.at(2)], &Integer.parse/1)
["0", "1"]

  
    
  
  Tuples


at also works with tuples:
iex> Deeply.get_all({"0", "1", "2"}, Lens.indices([0, 2]))
["0", "2"]
iex> Deeply.update({0, 1, 2}, Lens.at(2), & &1 * 1111)
{0, 1, 2222}
However, you cannot use an index out of range:
iex> Deeply.get_all({"0", "1"}, Lens.at(2))
** (ArgumentError) errors were found at the given arguments:

        * 1st argument: out of range
This is consistent with elem/2.
You also get an ArgumentError when attempting to put or update a value out of range.

  
    
  
  Enumerable types


Although Lens.at/1 is suggestive of Enum.at/2, you can't use it
with a non-list Enumerable. That makes sense for put and
update, since there's no general way to modify elements of an
Enumerable. Consider this:
iex> Deeply.put(0..5, Lens.at(1), 2)
?????
What would that even mean?
It seems you should be able to use Deeply.get_all, but you can't
because of
the lens implementation.

  
    
  
  Lenses specifically for adding to lists


Lens2.Lenses.Indexed supplies lenses that point, not to elements of a list,
but next to them. Consider Lens2.Lenses.Indexes.before/1:
iex> lens = Lens.before(2)
Given a list like [0, 1, 2], it lets you add a new element:
iex> Deeply.put(["0", "1", "2"], lens, :NEW)
["0", "1", :NEW, "2"]
It's rather peculiar to ask for a value at the place where there isn't
a value. If you do, you'll get a nil:
iex> Deeply.get_all(["0", "1", "2"], Lens.before(2))
[nil]
Similarly, you can use Deeply.update, but the update function will
always get a nil. Which is therefore a more elaborate version of
Deeply.put:
iex> Deeply.update(["0", "1", "2"], Lens.before(2), fn nil -> :NEW end)
[0, 1, :NEW, 2]

Non-list enumerables
    

Lens2.Lenses.Enum.all/0 is the building block for lens pipelines
that work with non-list Enumerable containers. 
To start the explanation, here it is applied to a keyword list:
iex> use Lens2
iex> Deeply.get_all([a: 1, b: 2], Lens.all)
[a: 1, b: 2]
Since a keyword list is just a list of key/value tuples, the output looks the same as the input. 
Here's the same example for a map:
iex> Deeply.get_all(%{a: 1, b: 2}, Lens.all)
[a: 1, b: 2]
Again, we get key-value pairs. To make that explicit:
iex> Deeply.get_all(%{1 => "1", 2 => "2"}, Lens.all)
[{1, "1"}, {2, "2"}]
This is the usual Elixir behavior when working with maps as enumerables:
iex> %{1 => "1", 2 => "2"} |> Enum.map(& &1)
[{1, "1"}, {2, "2"}]
We can pipe lens makers together to work on elements. Here's an
implementation of Lens2.Lenses.Keyed.map_values/0 that uses all and Lens2.Lenses.Indexed.at/1:
iex> Deeply.get_all(%{a: 1, b: 2}, Lens.all |> Lens.at(1))
[1, 2]

  
    
  
  Update


That's not quite map_values, though, because of update:
iex> Deeply.update(%{a: 1, b: 2}, Lens.map_values, & &1 * 111)
%{a: 111, b: 222}                 ^^^^^^^^^^^^^^^
iex> Deeply.update(%{a: 1, b: 2}, Lens.all |> Lens.at(1), & &1 * 111)
[a: 111, b: 222]                  ^^^^^^^^^^^^^^^^^^^^^^
all will always produce a List on update. The solution is to add
another lens that "pours" the list into a Collectable in a manner
analogous to Enum.into/2. Lens 1 provided Lens.into for that:
iex> map_values = Lens.all |> Lens.at(1) |> Lens.into(%{})
iex> Deeply.update(%{a: 1, b: 2}, map_values, & &1 * 111)
%{a: 111, b: 222}
That exists in Lens 2 as well. Experience shows, though, that Lens2.Lenses.Enum.into/2 is
error-prone. A pipeline like the above is a special case that's easy
to mis-generalize from. After writing a whole lot of text explaining
the issue (see here and
here), I realized that I ought to make the error harder to make. Hence Lens2.Lenses.Enum.update_into/2, which forces
you to make explicit which sub-pipeline produces the value to pour:
iex> map_values = Lens.update_into(%{}, Lens.all |> Lens.at(1))
                  ^^^^^^^^^^^^^^^^
iex> Deeply.update(%{a: 1, b: 2}, map_values, & &1 * 111)
%{a: 111, b: 222}
This matters when you're updating a non-list Enumerable nested within a container, like this one:
iex> container = [%{}, %{a: 1, b: 2}, %{}]
                       ^^^^^^^^^^^^^
iex> lens = Lens.at(1) |> Lens.update_into(%{}, map_values)
                          ^^^^^^^^^^^^^^^^^^^^^^^^^^^^^^^^^
iex> Deeply.update(container, lens, & &1 * 1111)
[%{}, %{a: 1111, b: 2222}, %{}]


  

  
    
    Information hiding - Lens 2 v0.1.0
    
    

    


  
  

    
Information hiding
    

I start from two beliefs:
	If you have complicated data, there's a reasonable chance your
first attempt at structuring it will be wrong. Or
adding new features will force restructuring.
This is as true for
functional languages (with their CRUD emphasis) as it is for OO
languages.  (After all, the languages where ideas about
modularity and information hiding took
root are at least arguably closer to modern functional languages
than to modern OO languages.)

	You want the inevitable restructuring to affect client code as
little as possible.
The fact that lenses are functions and
functional programming loves it some function composition suggests
that lenses make information hiding more idiomatic than Access
(which requires concatenating lists). 


My own practice (which is still evolving) is to give each struct their own lenses. Here's an example. 
I use TypedStruct to define
structs, because it's prettier than defstruct. Here's an example.
 defmodule Cluster do
   typedstruct enforce: true do
     field :name, atom
     field :downstream, %{atom => MapSet.t(atom)}
   end
   ...
It seems reasonable to have a downstream lens:
defmaker downstream, do: Lens.key!(:downstream)
I find this preferable to having client code flaunt its knowledge of
Cluster structure with code like this:
Deeply.get_only(cluster, Lens.key!(:downstream))
... or the equivalent:
get_in(cluster, [Access.key(:downstream)])
Code like this:
Deeply.get_only(cluster, Cluster.downstream)
... makes it less disruptive to change the Cluster
structure. Again: not a big deal with a shallowly nested container, but
lenses would be a waste of time if all structures were shallow and
predictable.
In fact, I like the CRUD abstraction enough that I've implemented shorthand
for cases where lens maker functions applied to structs take no arguments:
Deeply.get_only(cluster, :downstream)
After all, cluster contains its type, so the function call can be constructed easily enough.
There's something appealing to me about client code that says "within
a cluster, there's a downstream. Fetch that, no matter where it is."

In this style, containers that contain smaller containers will use the
smaller containers' lens makers to make their own lens makers. So suppose
we have a Network that contains Clusters:
 defmodule Network do
   typedstruct enforce: true do
     field :clusters_by_name, %{atom => Cluster.t}
     field :other_fields, any, default: :just_for_show
   end
   ...
Clients of Network will want to extract a particular named cluster's
downstream, which can be done like this:
defmaker downstream_of(cluster_name) do
  Lens.key!(:clusters_by_name) |> Lens.key!(cluster_name) |> Cluster.downstream
end
The important bit is that Network declines to make any guesses about
the structure of Cluster. It works from the lens API.

In the code I'm thinking of as I write, it turned out
that a Cluster will have not just one downstream list, but a
variety of them. You see, Clusters send "pulses" to their downstream
clusters. Originally, all pulses were the same, but later I needed to 
give pulses "types", so that there's a distinction between a
default pulse and a "control pulse". So suddenly, the downstream of
a cluster depends on the type of a pulse, so the code to retrieve the
downstream cluster names has to look like this:
Deeply.get_all(network, Network.downstream_of(originating_cluster_name, pulse_type))
The important thing here is that, once we have the cluster_names function, client
code needn't care whether clusters contain a map of pulse types to
downstream names, meaning the path would be described like this:
defmaker downstream_of(originating_name, pulse_type do 
  Lens.key!(:clusters_by_name)
  |> Lens.key!(originating_name)
  |> Lens.key(:downstream_by_pulse_type)
  |> Lens.key(pulse_type)
end
... or whether Network maintains a mapping from {cluster, pulse_type} to downstream names – removing from a Cluster any
notion of its "downstream" clusters. That would mean this path:
defmaker downstream_of(originating_name, pulse_type) do 
  Lens.key!(:downstreams)
  |> Lens.key!({originating_name, pulse_type})
end
The point is: having lens makers be module functions makes change easier.

Sample pipelines
    

Most lenses are straightforward, and so are most pipelines of
lenses. But lenses can be combined in some fairly complicated ways by
using lens makers in Lens2.Lenses.Combine.
Here I want to call out a few that
operate on other lenses and show examples of how they're used with
defmaker.
TO BE WRITTEN

Introduction to debugging and <code class="inline">def_raw_maker</code>
    

I really hate this damn machine.
I wish that they would sell it.
It never does quite what I want,
But only what I tell it.
– Traditional    

When using simple composed lenses like Lens.key(:a) |> Lens.keys([:aa, :bb]), you don't have to understand what's happening
behind the scenes. The lens descends through key :a, then through
keys :aa and :bb, and either returns a list of values or an
updated complete container (depending on whether you're using Deeply.get_all or Deeply.update. 
But sometimes – especially if you use some of the more unusual
combining lenses – you'll get surprised. I've reluctantly concluded
that the surest way to un-perplex yourself is to work through the
detailed steps of what's happening in the lens code. To do that, you
need to understand lenses well enough to write one from scratch. That's what this guide is about.
Note: you might want to defer reading this page until you actually
are perplexed or need to write a lens maker. Maybe you'll never need it!

A lens is an anonymous function created by a lens maker (a named
function). You can easily write a lens maker using def, but the
def_raw_maker macro handles some busywork for you. In either case, the
form of the lens function is fixed: you use a template and fill in
some blanks.
Because the template is somewhat conceptually tricksy, I'll work my
way up to its full glory by going through four versions:
	make a lens compatible with Deeply.get_all.
	make one compatible with Deeply.update. 
	combine the two into a lens function that works with both.
	add compatibility with Access (so lenses work with get_in, update_in, and friends).

Then I'll show how this understanding can be used to debug a couple of
lenses that have surprising behavior in pipelines. 
Finally, I'll describe the implementations of some interesting makers,
so that you can pattern your own makers off them, should you want your
own makers.

Continuation-passing style
    

It's easier to understand the implementation of lenses if you first
understand
"continuation-passing style". Lenses
don't exactly use continuation-passing style, but it's pretty close.
In this style, every function call is a two-part instruction:
	Dear function, do that thing you do, using these arguments, then...

	... do not return the result. Instead, pass it to the function I
also gave you in an argument. That function is called the
continuation (because it continues the overall computation of
which the function call is a part).


Any function can be converted to continuation-passing style. Here's a
continuation-passing version of Map.put/3:
def map_put(map, key, value, continuation) do
  Map.put(map, key, value)
  |> continuation.()
end
(You can find the code and tests quoted on this page at
continuation_passing_test.exs.)
And here's a use of map_put:
iex> map_put(%{}, :a, 1, & &1)
%{a: 1}
Here's a more elaborate call that puts two
different key/value pairs into a map:
iex> map_put(%{}, :a, 1,
             fn just_created_map ->
               map_put(just_created_map, :b, 2,
                       & &1)
             end)
%{a: 1, b: 2}
There are two continuations here: one that calls another instance of
map_put, and one that just returns the final value. The sequence of events is:
	map_put calculates %{a: 1}, and passes it to the bigger continuation.
	The continuation passes that value on to another map_put,
together with a second continuation, the identity function.
	The second map_put calculates %{a: 1, b: 2} and passes it to the second continuation.
	The second continuation just returns its value, so...
	... the second map_put returns the value to...
	... the first continuation, which returns it to...
	... the first instance of map_put which...
	... returns it to IEX for printing to the terminal.

(A sufficiently smart compiler could eliminate all but the last
return; indeed, continuation-passing style was invented for use in a
compiler as an intermediate form that lent itself to certain
optimizations.)
The code isn't what you'd call wildly readable. There's a general structure that's obscured by a bunch of constants. Here's the code with the constants marked:
  map_put(%{}, :a, 1,
  ^^^^^^^ ^^^  ^^  ^
          fn just_created_map ->
            map_put(just_created_map, :b, 2,
            ^^^^^^^                   ^^  ^
                    & &1)
                    ^^^^
          end)
Code full of constants can be turned into a function that takes appropriate arguments. Let's pull %{} and & &1out first:
two_puts =
  fn initial_value, final_continuation ->
     ^^^^^^^^^^^^^  ^^^^^^^^^^^^^^^^^^
    map_put(initial_value, :a, 1,
            ^^^^^^^^^^^^^
            fn just_created_map ->
              map_put(just_created_map, :b, 2,
                      final_continuation)
                      ^^^^^^^^^^^^^^^^^^
            end)
  end

iex> two_puts.(%{}, & &1)
%{a: 1, b: 2}
Next let's extract the two explicit calls to map_put into two "step"
arguments that give functions for the internal code to execute:
step_combiner =
  fn step1, step2 ->
     ^^^^^  ^^^^^
    fn initial_value, final_continuation ->
      step1.(initial_value,
      ^^^^^  fn just_created_map ->
               step2.(just_created_map,
               ^^^^^  final_continuation)
             end)
    end
  end
The step_combiner is a function that returns a function that
executes two computations in a row, passing the first result to the
second computation. It could be used like this:
two_puts =
  step_combiner.(
    fn map, continuation ->
      map_put(map, :a, 1, continuation)
    end,
    fn map, continuation ->
      map_put(map, :b, 2, continuation)
    end)

iex> two_puts.(%{}, & &1)
%{a: 1, b: 2}
Let's do a little cleanup and use named functions instead of anonymous functions. Let's make the steps with this function:
def make_put_fn(key, value) do
  fn map, continuation ->
    Map.put(map, key, value) |> continuation.()
  end
end
(I expanded out map_put because I won't be using it any more.)
Let's also make step_combiner a named function:
def step_combiner(step1, step2) do
  fn initial_value, final_continuation ->
    step1.(initial_value,
           fn step2_value ->
             step2.(step2_value,
                    final_continuation)
           end)
  end
end
Putting that together, we get:
iex> step1 = make_put_fn(:a, 1)
iex> step2 = make_put_fn(:b, 2)

iex> put_twice = step_combiner(step1, step2)
iex> put_twice.(%{}, & &1)
%{a: 1, b: 2}
Say, that looks kind of familiar:
iex> lens1 = Lens.key(:a)
iex> lens2 = Lens.key(:b)
iex> two_step = Lens.seq(lens1, lens2)
iex> Deeply.put(%{a: %{b: 2}}, two_step, :NEW)
%{a: %{b: :NEW}}
In fact, I can make it even more familiar. First, a function that
launches the combined step and supplies the identity function:
def do_to(structure, step) do
  step.(structure, & &1)
end
Second, a variant make_put_fn that takes a previous step and combines it with the one being created:
def make_put_fn(previous, key, value) do
  step_combiner(previous, make_put_fn(key, value))
end
And now:
iex> two_step = make_put_fn(:a, 1) |> make_put_fn(b: 2))
iex> do_to(%{c: 3}, two_step)
%{a: 1, b: 2, c: 3}

Version 1: <code class="inline">get_all</code>
    

Although the previous page showed that lenses have a family
resemblance to continuation-passing style, there are some differences:
	I had a single "launcher" function, do_to that is reminiscent of
Deeply.put. But lenses have several launcher functions for different
purposes. That is, a single lens can accommodate get_all, put, and
update functions.

	Put a bit differently, make_put_fn can only be used to put
values. What's being put is defined in the function that's like a
lens maker, rather than in the launcher function (like Deeply.put).


This page is a first step toward showing how lenses accomplish those
differences. It starts by adding just just a smidge onto the
continuation-passing style example.
Since there will be multiple implementations of lenses coming up, I'll
distinguish them by using a version number prefix instead of
Lens. Code on this page is version V1, and it will define
V1.at/1, V1.seq/2, and V2.all/0, as well as a Deeply-style
get_all function. The normal Deeply operations only work with the
V4 implementation, so each version will have its own operations
defined in a module I can't resist naming
Derply.
The code and tests for version 1 can be found in
implementation_v1_get_test.exs.

  
    
  
  V1.at


If we're using continuation-passing style as a model, a lens should
take a container as an argument, plus a continuation-ish function. I
say "continuation-ish" because, while the argument has the effect of
continuing a computation by descending more deeply into a container, a
proper continuation is the last thing a function does. A lens
function takes the return value of the continuation-ish function and does something with
it. So I'm going to call that argument a descender instead of a continuation. 
Here is the definition of V1.at/1:
def at(index) do
  fn container, descender ->
    gotten =
      Enum.at(container, index)
      |> descender.()
    [gotten]            # <<<<<
  end
end
The difference is that the descender's return value is wrapped in a list: 
that's the contract a lens must follow. There has to be a way to
distinguish between returning a list of values and a single value
that's a list. That's done by wrapping everything in a list, so that a
single value that's a list is returned like this:
[ [0, 1, 2] ]
... which is distinct from two values that are lists:
[ [0, 1, 2], [3, 4, 5] ]
... or six independent values (as you might get from Lens2.Lenses.Enum.all/0):
[ 0, 1, 2, 3, 4, 5 ]
The difference between V1 and continuation-passing style comes down
entirely to the little bit of code that executes after the descender.
(Note that this version of at only works with lists, whereas the
real one also works with tuples. Nothing informative about lenses
would be gained by dragging in tuples, so I won't.)

  
    
  
  Derply.get_all


As with the previous page's do_to function, get_all says what the last continuation in a chain should do. And that is... nothing: just return the value handed it back up the chain. 
So the implementation is simple:
def get_all(container, lens) do
  getter = & &1    # Just return the leaf value
  lens.(container, getter)
end
Now this works: 
iex> Derply.get_all(["0", "1"], V1.at(1))
["1"]

  
    
  
  V1.seq


As you've seen (I hope) a couple of times in this documentation, piping the lens from one
lens maker into another makes use of seq. That is, this:
Lens.at(1) |> Lens.at(2) 
... means that the second maker should produce code equivalent to this:
Lens.seq(Lens.at(1), Lens.at(2))
We can use the previous page's step_combiner as a template. Instead
of step1 and step2, there'll be outer_lens and inner_lens:
 def seq(outer_lens, inner_lens) do
   fn outer_container, inner_descender ->
     ...
   end
 end
The outer_descender has to be constructed by seq. I'll give it an explicit name:
 def seq(outer_lens, inner_lens) do
   fn outer_container, inner_descender ->
     outer_descender =                                 # <<<
      fn inner_container ->                            # <<<
        inner_lens.(inner_container, inner_descender)  # <<<
      end                                              # <<<
   ...
   end
 end
In our quasi continuation-passing style, seq has to do something
with the value returned by the outer_descender:
 def seq(outer_lens, inner_lens) do
   fn outer_container, inner_descender ->
     outer_descender =                                 
      fn inner_container ->                            
        inner_lens.(inner_container, inner_descender)  
      end                                              

    gotten =
      outer_lens.(outer_container, outer_descender)
    ????.(gotten)
    ^^^^^^^^^^^^^
   end
 end
What? Consider this container: [ [], [1, 2, 3] ] and the
pipeline from V1.at(1) to V1.at(2). 
	V1.at(1)'s lens first uses Enum.at(..., 1) to extract
[1, 2, 3], which it passes to the outer_descender and thus to
the lens from V1.at(2).

	V1.at(2)'s lens extracts 2 and passes it to the inner descender,
which immediately returns it.

	Now, the V1.at(2) lens wraps the result in a list, and returns it
to the V1.at(1) lens function. That function's descender returns it. 
At this point, we can represent the state of the V1.at(1) lens function as this:
 fn container = [ [], [1, 2, 3] ], descender ->
   gotten =
     # Enum.at(container, 1)
     # |> descender.()
     [2]
     ^^^
   [gotten]
 end

	As night follows day, the function will wrap [2] in a list, and
so return [[2]] to seq, which I'll represent as:
 fn outer_container, inner_descender ->
   outer_descender = ...

   gotten =
     # outer_lens.(outer_container, outer_descender)
     [[2]]
     ^^^^^
   ????(gotten)
 end
We've doubly-wrapped the return value. Returning it would violate the
lens contract. Unwrapping could be done in several ways, but this is
the right one:
 fn outer_container, inner_descender ->
   outer_descender = ...
   gotten = ...

   Enum.concat(gotten)
 end
Why that instead of, say, this:
   [gotten] =
     # outer_lens.(outer_container, outer_descender)

   gotten
Well...



  
    
  
  The garden of forking paths


With apologies to Jorge Luis Borges

The previous example was linear: the code descended to a single
"leaf" node, then returned the value found, wrapping and unwrapping as
needed. But lenses are built on the assumption that a single Deeply
operation may require the lenses to descend to a leaf, retreat to some
intermediate position in the container, descend again to another leaf,
retreat again, and so on.
Here's a simple example:
iex> nested = [ [0, 1, 2], [0, 1111, 2222]] 
iex> lens = V1.all |> V1.at(1)
iex> Derply.get_all(nested, lens)
[1, 1111]
The Enum.concat/1 call in seq is what produces that result. Let's
step through that, meaning I need to shgow you the code for V1.all. It's simple:
def all do
  fn container, descender ->
    for item <- container, do: descender.(item)
  end
end
In our example, all will do this, in effect:
    for item <- [ [0, 1, 2], [0, 1111, 2222]],
      do: descender.(item)
... which is the same as this:
    [
      descender.([0, 1, 2])
      descender.([0, 1111, 2222])
    ]
Since the descender calls at(1), that's equivalent to this:
    [
      [1],
      [1111]
    ]
... and that's why seq uses Enum.concat/1:
    iex> Enum.concat([ [1], [1111] ])
    [1, 1111]
The upshot of all this is that unless you're writing a special lens
maker like V1.seq, you won't have to worry about any unwrapping or
rewrapping. Just follow two rules:
	If you're fetching exactly one element, wrap it.
	If you're fetching zero to many elements, you've probably already got a list. Just return it. 

V1.at is an example of the first rule. V1.all is an example of the second.
Now let's implement Derply.update and lenses that will work with that.

Version 2: update
    

The code and tests for this version – version 2 – can be found in
implementation_v1_update_test.exs.
Update-capable lens code has the same "shape" as get-capable code. For
example, here's V2.update vs. V1.get_all. update takes an update
function and uses it as the final "descender":
def update(container, lens, update_fn) do      def get_all(container, lens) do
                            ^^^^^^^^^             
  lens.(container, update_fn)                    lens.(container, & &1)
                   ^^^^^^^^^                                      ^^^^
end                                            end
(A put operation is the same, except that it codes up a constant-returning update function:
def put(container, lens, constant) do
  lens.(container, fn _ -> constant end)
end
I won't mention put any more.)
It's worth emphasizing what happens in a call like:
iex> lens = Lens.key(:a) |> Lens.key(:b) |> Lens.key(:c)
iex> container = %{a: %{b: %{c: 1}}}
iex> update_fn = & &1 * 1111
iex> Deeply.update(container, lens, update_fn)
%{a: %{b: %{c: 1111}}}
The code descends all the way to the inner map %{c: 1}. It calls this function:
Map.update!(%{c: 1}, :c, update_fn)
... which produces %{c: 1111}. Having descended as far as it can, it
begins to "retreat" back to the original container. That means that
the lens for key(:b) will have pointers to two pieces of data:
%{b: %{c: 1}}  # embedded within the original container
%{c: 1111}     # a value returned from the lower lens
It must do this: 
Map.put(%{b: %{c: 1}}, :b, %{c: 1111})
... to make a new map %{b: %{c: 1111}}. And, retreating further up the pipeline, we
get this:
Map.put(%{a: %{b: %{c: 1}}}, :a, %{b: %{c: 1111}})
Every step of retreat "back up to" the original container allocates a
new map. That's just the way it is in a language without
mutability. (Such languages can make optimizations to share structure
between original and updated versions of structures, so long as no
user code can tell. I don't know if the Erlang virtual machine's
optimizations for maps would help with this example.)
With that background, here's the definition of V2.key:
def key(key) do
  fn container, descender ->
    updated =
      Map.get(container, key)
      |> descender.()

    Map.put(container, key, updated)
  end
end
In the leaf (%{c: 1}) case, that code has this effect:
    updated =
      Map.get(%{c: 1}, :c)
      |> (&1 * 1111).()
    Map.put(%{c: 1}, :c, updated)
In the next level up, the code should look like this:
    updated =
      Map.get(%{b: {c: 1}}, :b)
      |> leaf_descender.()
    Map.put(%{b: %{c: 1}}, :b, updated)
However, there's a V2.seq in between the key(:c) leaf lens and the
preceding key(:b) lens. What must that look like?
The V1 version of seq gets wrapped values and has to unwrap
them. But this V2 version gets a sub-container that's had a
repacement done. It doesn't have to do anything but return that value
to the previous lens, which will put it into the enclosing container.
So that's easy:
# `get_all` version                                     # `update` version
def seq(outer_lens, inner_lens) do                      def seq(outer_lens, inner_lens) do
  fn outer_container, inner_descender ->                  fn outer_container, inner_descender ->
    outer_descender =                                       outer_descender =
      fn inner_container ->                                   fn inner_container ->
        inner_lens.(inner_container, inner_descender)           inner_lens.(inner_container, inner_descender)
      end                                                     end

    gotten =                                                updated =
    ^^^^^^                                                  ^^^^^^^
      outer_lens.(outer_container, outer_descender)           outer_lens.(outer_container, outer_descender)

    Enum.concat(gotten)                                     updated
    ^^^^^^^^^^^^^^^^^^^                                     ^^^^^^^
  end                                                     end
end                                                     end
When it comes to ordinary lenses, the changes are equally trivial. Here's at:
# `get_all` version                # `update` version
def at(index) do                   def at(index) do
  fn container, descender ->        fn container, descender ->
    gotten =                          updated =
      Enum.at(container, index)         Enum.at(container, index)
      |> descender.()                   |> descender.()
    [gotten]                          List.replace_at(container, index, updated)
    ^^^^^^^^                          ^^^^^^^^^^^^^^^^^^^^^^^^^^^^^^^^^^^^^^^^^^
  end                               end
end                               end
all doesn't have to change at all:
def all do                          def all do
  fn container, descender ->          fn container, descender ->
    for item <- container,              for item <- container,
        do: descender.(item)                do: descender.(item)
  end                                 end
end                                 end
Next is to combine the V1 and V2 lenses into a template that works for both getting and updating. 

Version 3: <code class="inline">get_and_update</code>
    

The code and tests for this version can be found in
implementation_v3_get_and_update_test.exs.
Lens2.Deeply.get_and_update/3 is analogous to Elixir'sbuilt-in
get_and_update_in/3. As with get_and_update_in/3, it takes a function
that returns both an original value and its updated version:
iex> tuple_returner = fn value -> {value, inspect(value)} end
iex> Deeply.get_and_update(%{a: 1}, Lens.key(:a), tuple_returner)
{[1], %{a: "1"}}
To implement this:
	The descender function will, as before, take as its argument some
container. But it will return a {gotten, updated} tuple.

	A lens function will also return such a tuple to its caller.

	Derply.get_and_update can just return the tuple:
 def get_and_update(container, lens, tuple_returner) do
   lens.(container, tuple_returner)
 end


Here is a new version of at. It gets a value, sends it to the
descender, and processes the two returned elements as in version 1
and version 2 (respectively):
def at(index) do
  fn container, descender ->
    {gotten, updated} =
     ^^^^^^  ^^^^^^^
      Enum.at(container, index)
      |> descender.()

    {[gotten], List.replace_at(container, index, updated)}
     ^^^^^^^^  ^^^^^^^^^^^^^^^^^^^^^^^^^^^^^^^^^^^^^^^^^^
  end
end
Not a very big change. The same can be done for seq:
def seq(outer_lens, inner_lens) do
  fn outer_container, inner_descender ->
    outer_descender =
      fn inner_container ->
        inner_lens.(inner_container, inner_descender)
      end

    {gotten, updated} =
     ^^^^^^  ^^^^^^^
      outer_lens.(outer_container, outer_descender)

    {Enum.concat(gotten), updated}
     ^^^^^^^^^^^^^^^^^^^  ^^^^^^^
  end
end

  
    
  
  The other Derply operations


Derply.update/3 uses a function that returns an updated container
(not a {gotten, updated} tuple). By converting that update function
into a "tuple returner", it can just use Derply.get_and_update and
ignore the gotten value:
def update(container, lens, update_fn) do
  tuple_returner = & {&1, update_fn.(&1)}
                   ^^^^^^^^^^^^^^^^^^^^^^
  {_, updated} = get_and_update(container, lens, tuple_returner)
                                                 ^^^^^^^^^^^^^^
  updated
end
get_all also calls get_and_update; it just returns the gotten
tuple element instead of updated. But there's a problem: What
tuple-returning function does it call?
def get_all(container, lens) do
  tuple_returner = & {&1, ?????????}
                          ^^^^^^^^^
  {gotten, _} = get_and_update(container, lens, tuple_returner)
  gotten
end
There's a sense in which it doesn't matter, because whatever update
happens is thrown away. So it uses the most innocuous function
possible: the identity function:
  tuple_returner = & {&1, &1}

  
    
  
  Record scratch noise


But wait. The at lens's return value is constructed like this:
  {gotten, updated} = ...

  {[gotten],
   List.replace_at(container, index, updated)  ### ‽‽‽‽‽
  }
Isn't List.replace_at/3 going to construct a new list, one that's
identical to the old list?  And in a pipeline of lenses, won't there
be a whole series of lenses doing that – going to a lot of work to
construct a container that's == to the original container,
allocating memory like mad? Only to throw away the newly-constructed
container?
Well, yes. In the case of at, that's exactly what
happens. Fortunately, composite structures like Map, structs, and
MapSet are clever enough not to allocate a new structure if they see
you're trying to put the same value that's already there. Instead,
they just return the original structure.
Still, for a large list (and other not-so-clever datetypes), there's
potentially a lot of wasted work. That means that Deeply.get might
get progressively slower than get_in as containers get bigger. 
I did some extremely crude benchmarking. For maps and structs, are
about twice as slow as Access functions, and the relationship seems
constant(ish). (I used the real Lens code, not V3.) For lists,
lenses get slower and slower as the lists get longer, but there's not
an alarming "knee" in a graph of 4,000,000 repetitions of probes into
lists of sizes up to 10,000 elements:
[image: alt text coming]
The ancient advice applies: let profiling tell you where your problems are.
(Access avoids the problem of pointless updates by having two paths
through the equivalent of lens functions: one that's purely a "get"
and one that's a "get and update". That means it's more work to add a
new data structure to Access than it is to write a lens for it, but you're more likely to need to replace a lens pipeline with a hand-coded get or update function.)
Speaking of Access...

Version 4: compatibility with <code class="inline">Access</code>
    

The code and tests for this version can be found in
implementation_v4_access_test.exs.
We want lenses to be compatible with get_in/2 and friends:
iex> tuple_returner = & {&1, inspect(&1)}
iex> lens = Lens.at(1)
iex> get_and_update_in([0, 1, 2], [lens], tuple_returner)
                                  ^^^^^^
{[1], [0, "1", 2]}
... including as only part of a list argument:
iex> container = [0, %{a: 1}, 2]
iex> get_and_update_in(container, [lens, :a], tuple_returner)
                                  ^^^^^^^^^^
{[1], [0, %{a: "1"}, 2]}
In fact, let's just define our Derply functions in terms of the Elixir built-ins:
def get_and_update(container, lens, tuple_returner) do
  Kernel.get_and_update_in(container, [lens], tuple_returner)
end

def update(container, lens, update_fn) do
  Kernel.update_in(container, [lens], update_fn)
end

def get_all(container, lens) do
  Kernel.get_in(container, [lens])
end
Actually, we don't need to define Derply functions at all, as these
are the actual definitions used in Lens2.Deeply.

  
    
  
  The behaviour


A function suitable for Access is must have the following interface:
      fn
        :get, container, continuation ->
           ...

        :get_and_update, container, tuple_returner ->
           ...
      end
The :get and :get_and_update arguments are required to distinguish
the two branches because the tyhpes of the second and third arguments
are the same in both functions. The container can be any type, and
both continuation and tuple_returner are arity-one
functions. Let's look at the body of the :get case first:
        :get, container, continuation ->
          {gotten, _} = lens.(container, &{&1, &1})
          continuation.(gotten)
Except for the the use of the continuation argument, this does the
same thing as the V3 version of get_all. In particular it uses the
tuple-returner &{&1, &1} to produce the throw-away version of the
updated container. (So the wasteful multi-level allocation happens with
get_in/2.)
The continuation represents what comes after this function in a call to get_in. So, in this:
get_in(..., [Lens.at(0), :a, :b])
... the continuation is a function representing the descent through
the keys :a and :b. Because Deeply.get_all uses a singleton
list:
def get_all(container, lens) do
  Kernel.get_in(container, [lens])
                           ^^^^^^
end
... there's nothing more to do, so the continuation will be our old
friend & &1. That is, when it comes to lenses, we don't need to
worry about it.
The :get_and_update clause looks just like the code for version 3's Derply.get_and_update:
        :get_and_update, container, tuple_returner ->
          lens.(container, tuple_returner)
The code for the actual lenses is the same as in version 3, except
that it should use the macro Lens2.Makers.def_raw_maker/2 instead of
def. def_raw_maker wraps its body in a :get/:get_and_update
function, and also arranges to create the lens maker with an
additional argument – the one used in a pipeline. So, this use of def_raw_maker:
def_raw_maker at(index) do
  fn container, descender -> ... end
end
... will expand out to:
def at(index) do 
  lens = fn container, descender -> ... end         # <<<<<

   fn
     :get, container, continuation -> 
       {gotten, _} = lens.(container, &{&1, &1})
       continuation.(gotten)

     :get_and_update, container, tuple_returner ->
       lens.(container, tuple_returner)
   end
end

def at(previously, index) do 
  Lens.seq(previous, at(index))
end

Example: <code class="inline">Lens.into</code>

Example: <code class="inline">Lens.const</code>

Some exemplary implementations

DRAFT: Popping
    

Access supports "popping" elements from containers. There are two variants.
	The first variant allows the tuple-returner passed to
get_and_update_in to return a :pop value instead of a tuple. That informs
the calling code to remove an element. Here's a rather contrived example. It pops
a key if its value is negative, stringifies it otherwise:
 iex> pop_negative = 
 ...>  & if &1 < 0, do: :pop, else: {&1, inspect(&1)}
 iex> container = [%{a: 1}, %{a: -1}]
 iex> get_and_update_in(container, [Access.all, :a], pop_negative)
 {[1, -1], [%{a: "1"}, %{}]}

	The second variant pops particular keys, regardless of their values:
 iex> container = [ %{a: %{aa: 1, bb: 2}},
                    %{a: %{aa: 11, bb: 22}}]
 iex> pop_in(container, [Access.all, :a, :aa])
 {[1, 11], [%{a: %{bb: 2}}, %{a: %{bb: 22}}]}



  
    
  
  The problem with get_and_update


Access works on a few data types and requires them each to implement
get_and_update. It's that data-structure-specific function that
handles :pop return values. Here, for example is Map.get_and_update/3
   def get_and_update(map, key, tuple_returner)
     current = get(map, key)

     case tuple_returner.(current) do
       {gotten, updated} ->
         {gotten, put(map, key, updated)}

       :pop ->
         {current, delete(map, key)}
     end
   end
We could not put a similar :pop clause in the function that
def_raw_maker wraps around a lens function because it's too late. The
wrapper takes control after the lens (and descender) have modified
its container.  So a :pop case for a map would be working with an
updated value like %{a: :pop}, not a simple :pop-or-tuple return
value. Whereas Map.get_and_update/3 has to work with exactly one
datatype, the generic wrapper would have to work with
everything. (And, given lenses like
Lens2.Lenses.Combine.repeatedly/1, it couldn't even just scan the
top level for :pop.)
So supporting this kind of popping would mean we'd have to change quite a number
of lenses: 
def_raw_maker key(key) do
  fn container, descender ->
    {gotten, updated} = descender.(Map.get(container, key))
    {[gotten], Map.put(container, key, updated)}
  end
end
... to this:
def_raw_maker key(key) do
  fn container, descender ->

    current = Map.get(container, key)
    case descender.(current) do
      :pop ->
        {[current], Map.delete(container, key)}
      {gotten, updated} ->
        {[gotten], Map.put(container, key, updated)}
    end
  end
end
The same would have to be done for key?/1, key!/1, at/1. So it's
not surprising the original author of Lens 1 left it out.
... though, now that I think about it, it wouldn't be so hard.

  
    
  
  The problem with pop_in


You can also directly instruct Access-compatible types to pop an element at a named key:
iex> container = %{a: [0, 1, 2]}
iex> pop_in(container, [:a, Access.at(1)])
{1, %{a: [0, 2]}}
In this case, the accessor list contains a function (Access.at(1)),
so it's handled as in the get_and_access_in. Effectively, it's turned into:
iex> always_pop = fn _ -> :pop end
iex> get_and_update_in(container, [:a, Access.at(0)], always_pop)
{0, %{a: [1, 2]}}
In the more common case, where all the access list elements are keys,
it's handled by the Access-compatible datatype's pop(container, key) function, applied to the final container and key. So, in lens terminology,
the final descender is (for a Map):
descender = fn Map.pop(%{a: "pop", b: "no"}, :a) end
The Access code can do that because the access list is a list, so
it can inspect the last element as it recursively descends the
container. The actual code pattern matches on a single-element
list. See the fifth and sixth lines below (from the Access.pop/2 source):
defp pop_in_data(data, [fun])        when is_function(fun),
  ...
defp pop_in_data(data, [fun | tail]) when is_function(fun),
  fun.(:get_and_update, data, fn _ -> :pop end)
defp pop_in_data(data, [key]),                  # <<<<
  do: Access.pop(data, key)                     # <<<<
defp pop_in_data(data, [key | tail]),
  ...
Lens code operates on functions, not lists of keys and functions, so
it can't know when it's descended to the final element. Again, any
implementation of Deeply.pop_in would devolve to changing lenses to
handle a :pop result returned from a descender. 

DRAFT: Why <code class="inline">Lens.into</code> invites bugs
    

Here I'll describe a very common mistake using
Lens2.Lenses.Enum.into/2. This is sort of a digression, but it
reinforces some ideas from the previous two pages.
Lens2.Lenses.Enum.all/0 turns a pointer to an Enumerable into a set of
pointers to all its elements. It's most often used with lists, but it
doesn't have to be.
 iex> use Lens2
 iex> Deeply.get_all(1..5, Lens.all)
 [1, 2, 3, 4, 5]
Suppose we want to increment each of the numbers. That doesn't
actually make sense for a range, but let's see what happens:
iex> Deeply.update(1..5, Lens.all, & &1+1)
[2, 3, 4, 5, 6]
Or just overwrite all the values:
iex> Deeply.put(1..5, Lens.all, 1111)
[1111, 1111, 1111, 1111, 1111]
For any update operation, Lens.all/0 produces a list. Suppose we
instead want a MapSet. We could do that with Enum.into/2:
iex> Deeply.put(1..5, Lens.all, 1111) |> Enum.into(MapSet.new)
MapSet.new([1111])
(Notice that collapsed all the 11111 values into one, because
MapSets don't allow duplicates. Maybe that's why we wanted a MapSet.)
There is, however, a lens that is the equivalent of Enum.into/2:
Lens2.Lenses.Enum.into/2:
iex> Deeply.put(1..5, Lens.all |> Lens.into(MapSet.new), 11111)
MapSet.new([11111])
Looks good. Let's even put it in a module a a predefined lens maker:
defmodule MyLenses do
  defmaker as_mapset,
    do: Lens.all |> Lens.into(MapSet.new)
end
Now it happens that we have a structure containing a range, and we
want to increment the range values into a mapset. Seems easy:
iex> lens = Lens.key(:a) |> MyLenses.as_mapset
iex> Deeply.update(%{a: 1..5}, lens, & &1+1)
%{a: MapSet.new([2, 3, 4, 5, 6])}
That looks good. Time passes, and you come across a similar situation. This time you decide a named lens maker is overkill. You'll just pipe the lenses together at the point of use:
iex> lens = Lens.key(:a) |> Lens.all |> Lens.into(MapSet.new)
iex> Deeply.update(%{a: 1..5}, lens, & &1+1)
MapSet.new([a: [2, 3, 4, 5, 6]])
Look closely at that: it's not a map containing a mapset. It's a
mapset containing a keyword list.
Two questions: 
	What went wrong with this?
 Lens.key(:a) |> Lens.all |> Lens.into(MapSet.new)

	And why did this work...?
 lens = Lens.key(:a) |> MyLenses.as_mapset
...given that as_mapset is just:
 Lens.all |> Lens.into(MapSet.new)



  
    
  
  Why the pipeline fails


What if we manually expand out the pipeline into a nested pair of Lens.seq?
Lens.seq(Lens.key(:a), Lens.seq(Lens.all, Lens.into(MapSet.new)))
** (UndefinedFunctionError) function Lens2.Lenses.into/1 is undefined or private. Did you mean:

  * into/2
  * into/3
Lens.all |>

  
    
  
  Why the named function doesn't fail

Lens2 
    



      
Use this module for convenience.
A module that does that can...
	... access all the lens-making functions from the Lens 1 package
under the same names: Lens.key/1, for example. See
Lens2.Lenses for the complete list.

	... make lenses for MapSet containers with Lens2.MapSet and
lenses for BiMap with Lens2.BiMap. Those are aliases for
Lens2.Lenses.MapSet and Lens2.Lenses.BiMap.

	... traverse containers with lens-using functions like
Deeply.update. (See Lens2.Deeply.)

	... define its own lens makers with def_raw_maker and defmaker. (See
Lens2.Makers.)


It also defines the types used in specs.

Lens2.Deeply 
    



      
Operations that work with lenses. The API is close to the familiar get, put, update one.
Note, though, these differences:
	Most APIs with get (like Map and Keyword) return a
single value from a container. Lenses point at zero or more
places within a container, so the natural "read" operation is to
return a collection (specifically, a List). Naming that
operation get invites mistakes, so the fundamental "read"
operation is Lens2.Deeply.get_all/2. In the case where you know there's
a single value being returned, you can use Lens2.Deeply.get_only/2 to avoid having
to pick a value out of a singleton list.

	In those Elixir core APIs, it's the operations that decides what to do about
missing keys and the like. Consider Map.update/4 (use a default)
and Map.update!/3 (raise an error). In this lens package, it's the
lens that decides. See, for example, Lens2.Lenses.Keyed.key/1,
Lens2.Lenses.Keyed.key?/1, and Lens2.Lenses.Keyed.key!/1. Or see this page in the tutorial.

	When you have structs, you can add lenses to the module interface so that you can
write code like:
 Deeply.put(%Container{...}, Container.place(...), ...)
In the reasonably-common case where the module's lens function takes no argument, you
can just use the bare atom name:
 Deeply.put(%Container{...}, :place, ...)
See this page for more.


The functions in this module have simple implementations because lenses are
compatible with the Access behaviour. For example,
Lens2.Deeply.put/3 is just a wrapper around a call to
put_in/3:
 def put(container, lens, value),
     do: put_in(container, [lens], value)
You can use the Kernel functions if you prefer. Don't forget the bracket!
For convenience and backwards compatibility, this package also provides the Lens 1 functions to_list/2, and one!/2.

Lens2.Makers 
    



      
Two ways of defining named lens-making functions that improve on def.
Lens makers are named functions that, when called, return a
 lens. Lens makers are created in three ways:
	Using plain old def:
def leaf(key), do: Lens.key(:down) |> Lens.key(key)
This is fine, but you cannot use leaf itself in a pipeline. That is, this
will fail to compile:
Lens.key(:upper) |> leaf(:bottom)
error: undefined function leaf/2

	Because of that, it's better to use defmaker/2 (alternately, deflens/2).
This creates a lens maker from one or more other lens makers, most often
by piping them into each other. For example, a lens maker for two-level maps
could be defined like this:
defmaker leaf(key), do: Lens.key(:down) |> Lens.key(key)
# Works fine:
Lens.key(:upper) |> leaf(:bottom)

	def_raw_maker/2 (alternately, deflens_raw/2, is used for cases where
a lens maker can't be made from other lens makers. For example, the definition of
Lens2.Lenses.Keyed.key!/1 looks like this:
@spec key!(any) :: Lens2.lens
def_raw_maker key!(key) do
  fn composed, descender ->
    {gotten, updated} = descender.(DefOps.fetch!(composed, key))
    {[gotten], DefOps.put(composed, key, updated)}
  end
end

Lens2.Lenses 
    



      
Aggregates all the lenses in
Lens2.Lenses.Combine,
Lens2.Lenses.Enum,
Lens2.Lenses.Filter,
Lens2.Lenses.Indexed, and
Lens2.Lenses.Keyed.
Traditionally, this module is aliased to Lens, so that the makers
have the same name as in the Lens 1 package. See the Lens2 module.

Lens2.Lenses.BiMap 
    



      
Lens makers for the BiMap
bidirectional map package.
A BiMap allows a "reverse lookup": using a value to find the corresponding key.
iex>  bimap = BiMap.new(%{1 => "1111", 2 => "2222"})
iex>  BiMap.get(bimap, 1)
"1111"
iex>  BiMap.get_key(bimap, "1111")
1
A BiMap differs importantly from a map in that it requires
values, not just keys, to be unique. This can cause some confusion
as BiMap's put operation can cause existing bindings to disappear.
 iex>  bimap = BiMap.new(a: 5, b: 6)
 iex>  BiMap.put(bimap, :b, 5)
 BiMap.new(b: 5)   # where did `:a` go?
I've found the problem gets worse with lenses (because you may be
putting multiple places in a single call, and the place you want
to change may be nested below the BiMap, making it harder to
notice the constraint is relevant.) You might prefer
BiMultiMap (which comes with BiMap).
Note that BiMap does not implement Access. However, lenses
created with these functions do, so they can be used with get_in/2
and friends.
This module follows the convention of using names like key to mean
using a key to obtain a value. Names like to_key go in the reverse
direction: from value to key. Using value(1) to refer to the key
:a would be just too weird.

Lens2.Lenses.Combine 
    



      
Lenses that combine lenses to get new lenses.

Lens2.Lenses.Enum 
    



      
Lenses that work on Enumerable and Collectable containers.

Lens2.Lenses.Filter 
    



      
Lenses that reduce a set of pointers into a smaller set of pointers.
Consider a use of, say, Enum.filter:
iex> [-2, -1, 0, 1, 2] |> Enum.filter(& &1 > 0)
[1, 2]
Functions in this module convert such a function into a lens that
will (eventually) apply the same filter to already-selected elements
of a container.
iex> lens = Lens.keys([:a, :c]) |> Lens.filter(& &1 > 0)
iex> container = %{a: 1, b: 9, c: -1}
iex> Deeply.get_all(container, lens)
[1]
iex> Deeply.update(container, lens, & -1111 * &1)
%{a: -1111, b: 9, c: -1}

Lens2.Lenses.Indexed 
    



      
Lenses specific to lists, plus one that works on both lists and tuples.
Deeply.put/3 and Deeply.update/3 produce lists when applied to
lists, tuples when applied to tuples. As always, Deeply.get_all/2
produces a list whether it operates on a list or a tuple.
These lenses do not work on Enumerables (except for lists).

Lens2.Lenses.Keyed 
    



      
Lenses helpful for working with structs, maps, and types implementing the Access behaviour.
Unlike Access, these functions make no distinction between structs
and lenses. All operate on both. Lens2.Deeply.put/3 and
Lens2.Deeply.update/3 will return a plain map if given one. If
given a struct, they will return a struct of the same type.
These lenses are available under the Lens alias when you use Lens2.
These lenses can be used on Keyword lists, but
the results may surprise you.
	Operations apply only to the first matching key:
 iex>  use Lens2
 iex>  keylist = [a: 1, other: 2, a: 3]
 iex>  Deeply.get_all(keylist, Lens.key(:a))
 [1]         # not [1, 3]
 iex>  Deeply.get_all(keylist, Lens.keys([:a, :other]))
 [1, 2]      # not [1, 2, 3]

	Update operations will produce maps rather than keyword lists.


See the Lens2.Lenses.Keyword module for an alternative.

Lens2.Lenses.Keyword 
    



      
A variant of Lens2.Lenses.Keyed that applies to all matching Keyword keys.
The Kernel "_in" functions apply only to the first matching key in a Keyword list:
iex(3)> keylist = [a: 1, other: 2, a: 3]
[a: 1, other: 2, a: 3]
iex(4)> get_in(keylist, [:a])
1
iex(5)> put_in(keylist, [:a], 33)
[a: 33, other: 2, a: 3]
iex(6)> update_in(keylist, [:a], & &1 + 100)
[a: 101, other: 2, a: 3]
Lenses like Lens2.Lenses.Keyed.key/1 mimic that behavior. (Indeed,
they're implemented using it.) Given that the Joe Q. Average lens
points at multiple places, that's counterintuitive. The lenses in this
module will get, put, and update all matching keys.
Problems: duplicate keys.
Some set-type operations don't produce a keyword list.
Deeply.put([a: 1], Lens.map_values, :NEW)
%{NEW: 1}

Lens2.Lenses.MapSet 
    



      
Lenses that work with MapSet structures.

mix has.smart.put 
    



      
Is the data type smart about "put" operations that produce identical structures?
A naive implementation of Map would be such that:
iex> Map.get(map, :key)
:some_value
iex> Map.put(map, :key, :some_value)
... would allocate a completely new map. Maps and structs are, in fact, smarter than
that. put just returns the original map.
This matters for the current implementation of lenses because
Deeply.get_all will do such equality-preserving puts (and then
throw the result away). Naive implementations of a container data
type will do extra work that get_in avoids.
If you're implementing lenses for a new data structure, you nmight want to know
whether it's smart or naive.
$ mix has.smart.put
Is the data structure smart enough to not create an identical copy?
List: false
Map: true
Struct: true
MapSet: true


      




  

  
    
    mix timings.keyed - Lens 2 v0.1.0
    
    

    


  
  

    
mix timings.keyed 
    



      
Crude timings that compare Access and Lens operations on a struct+map container.
The container has this structure:
 .Network{
   cluster_count: 10,
   names_by_index: %{
     1 => :name1,
     2 => :name2,
     3 => :name3,
     ...
   },
   clusters_by_name: %{
     name1: %Mix.Tasks.Timings.Keyed.Cluster{
       router: %{c: 0, a: 0, d: 0, b: 0, e: 0},
       fun: &Kernel.inspect/1,
       string: ":name1",
       atom1: :name1
     },
     name2: %Mix.Tasks.Timings.Keyed.Cluster{
       router: %{c: 0, a: 0, d: 0, b: 0, e: 0},
       fun: &Kernel.inspect/1,
       string: ":name2",
       atom1: :name2
     },
     ...
   }
 }
Get operations (get_in, Deeply.get_all) extract values like
 network.clusters_by_name[:name2].router.[:c]
Update operations increment those values.
Call with a list of the number of clusters:
  % mix timings.keyed 10 100 1000
Each number creates a scaled network and runs the operation 40_000_000 times.
I wrote this because I realized that Deeply.get_all will create a
copy of the original structure as it "retreats" from finding the
gotten values. Fortunately, for maps and structs,
Map.put(container, key, new_value) actually returns the original
container if the new value is the same as the existing value, so
that's not so bad. But it's interesting to compare Access to Lens anyway.

mix timings.list 
    



      
Crude timings that compare Access and Lens operations on a struct+map container.
The container has this structure:
 .Network{
   cluster_count: 10,
   names_by_index: %{
     1 => :name1,
     2 => :name2,
     3 => :name3,
     ...
   },
   clusters_by_name: %{
     name1: %Mix.Tasks.Timings.Keyed.Cluster{
       router: %{c: 0, a: 0, d: 0, b: 0, e: 0},
       fun: &Kernel.inspect/1,
       string: ":name1",
       atom1: :name1
     },
     name2: %Mix.Tasks.Timings.Keyed.Cluster{
       router: %{c: 0, a: 0, d: 0, b: 0, e: 0},
       fun: &Kernel.inspect/1,
       string: ":name2",
       atom1: :name2
     },
     ...
   }
 }
Get operations (get_in, Deeply.get_all) extract values like
 network.clusters_by_name[:name2].router.[:c]
Update operations increment those values.
Call with a list of the number of clusters:
  % mix timings.keyed 10 100 1000
Each number creates a scaled network and runs the operation 40_000_000 times.
I wrote this because I realized that Deeply.get_all will create a
copy of the original structure as it "retreats" from finding the
gotten values. Fortunately, for maps and structs,
Map.put(container, key, new_value) actually returns the original
container if the new value is the same as the existing value, so
that's not so bad. But it's interesting to compare Access to Lens anyway.

